Shader "HolaShader/BlinePhongFrag"
{
	Properties 
	{
		_Diffuse ("Diffuse", Color) = (1, 1, 1, 1)
		_Specular ("Specular", Color) = (1, 1, 1, 1)
		_Gloss ("Gloss", Range(8.0, 256)) = 20
	}

	SubShader 
	{
		Pass
		{
			//声明该Pass的光照模式
			Tags {"LightMode" = "ForwardBase"}

			CGPROGRAM

			#pragma vertex vert
			#pragma fragment frag

			#include "Lighting.cginc"

			fixed4 _Diffuse;
			fixed4 _Specular;
			float _Gloss;

			struct a2v 
			{
				float4 vertex : POSITION;
				float3 normal : NORMAL;
			};

			struct v2f
			{
				float4 pos : SV_POSITION;
				float3 worldNormal : TEXCOORD0;
				float3 worldPos : TEXCOORD1;
			};

			v2f vert (a2v v) 
			{
				v2f o;
				//转换顶点空间：模型=>世界
				o.worldPos = mul(unity_ObjectToWorld, v.vertex).xyz;
				//转换顶点空间：模型=>裁剪
				o.pos = UnityObjectToClipPos(v.vertex);

				//转换法线空间：模型=>世界
				o.worldNormal = mul(v.normal, (float3x3)unity_WorldToObject);
				  
				return o;
			}

			fixed4 frag(v2f i) : SV_Target
			{
				//获取环境光
				fixed3 ambient = UNITY_LIGHTMODEL_AMBIENT.xyz;

				//获取法线信息
				fixed3 worldNormal = normalize(i.worldNormal);

				//获取光线方向
				fixed3 worldLightDir = normalize(_WorldSpaceLightPos0.xyz);

				//计算漫反射信息
				fixed3 diffuse = _LightColor0.rgb * _Diffuse.rgb * saturate(dot(worldNormal, worldLightDir));

				//获取世界空间视角方向
				fixed3 viewDir = normalize(_WorldSpaceCameraPos.xyz - i.worldPos.xyz);

				//计算Half方向
				fixed3 halfDir = normalize(worldLightDir + viewDir);

				//计算高光信息
				fixed3 specular = _LightColor0.rgb * _Specular.rgb * pow(saturate(dot(worldNormal, halfDir)), _Gloss);

				return fixed4(ambient + diffuse + specular, 1.0);

			}

			ENDCG
		}
	}
	FallBack "Specular"

}